

# Maple to Accompany Differential Equations

Douglas Hundley  
Department of Mathematics and Statistics  
Whitman College

August 27, 2018



# Contents

<b>1</b>	<b>Getting Started</b>	<b>5</b>
1.1	Getting Started . . . . .	5
1.2	Algebra . . . . .	6
1.3	Reading Questions . . . . .	9
1.4	Plotting . . . . .	10
1.5	Calculus . . . . .	11
<b>2</b>	<b>Working with Worksheets</b>	<b>13</b>
2.1	Adding text to a Maple session . . . . .	13
2.2	Saving your work . . . . .	13
2.3	Printing . . . . .	14
<b>3</b>	<b>Lab 1: Slope Fields</b>	<b>15</b>
3.1	Lab 1 Questions . . . . .	16

These notes are modifications of an older manuscript authored by Prof. David Guichard and Prof. Albert Schueller.



# Chapter 1

## Getting Started

### 1.1 Getting Started

Maple is a powerful mathematical calculator, often called a computer algebra system (CAS) or symbolic mathematics program, because unlike most calculators it can do algebra and calculus symbolically. For example, it can expand a product, factor a polynomial, compute the derivative or compute an integral.

The strength of a CAS is that the computations are done exactly, with no numerical error. If a numerical estimation is desired, it is also possible to have Maple compute that (and sometimes the only meaningful answer is a numerical estimation). Finally, Maple can do some quite sophisticated graphing in both 2 and 3 dimensions, and that is a powerful reason to become familiar with a CAS.

This “tutorial” goes through some of the features of Maple. You should type in the Maple commands given on the left- On the right are comments and explanations about what those commands do.

Maple is preloaded on all computers in the Math and CS departments. You may also install Maple on your home computer (see your instructor for the details).

#### Setting up the Default Login

Once Maple has started, it is useful to set some options so that they are the default options:

- We will always work in **Worksheet** mode, NOT in **Document** mode. You can tell that you are in **Worksheet** mode because you will see the red cursor: >

When the Maple program begins, select **Start with a Blank Worksheet**.

- To make some options default, go to: **Tools**, then **Options**
  - Go to the **Display** tab, and set **Input display:** to **Maple Notation**
  - Go to the **Interface** tab, and choose: **Default format for new worksheets:**, and select **Worksheet**
- Select the **Apply Globally** button at the bottom.
- You might close, then re-start Maple to see that the options took (that is, Maple should automatically start in **Worksheet** mode).

- Hint: You can shut off the somewhat annoying startup dialog box once you get the defaults set- Uncheck the little box in the lower left corner: Show this dialog on startup.
- Hint: You can close all the buttons on the left for the time being- We don't want to use the shortcut keys until we understand what is behind them. Use the small triangle key on the divider to shut them off temporarily.

## 1.2 Algebra

As we noted previously, Maple can perform exact arithmetic and algebra. Here are some examples. Again, you should type in the commands given on the left. You should make a note of what Maple is returning, and some comments are given on the right.

### 1.2.1 Numbers

```
2+3;
123456789*987654321;
```

You must type the semicolon at the end of the line to signal the end of your expression.

```
2+3
34*98
-45;
```

Type this as is, without the semicolon on the first line. Maple will warn you about the missing semicolon, but will do the calculation anyway. Sometimes you may want to start a new line as part of the same expression; you can do this by holding the shift key while you press return. Do this after the "98".

```
2/5 + 1/3;
23/45-167/345*2/3;
(23/45)-(167/345)*(2/3);
```

All rational arithmetic is done exactly. The usual order of operations apply, but this is bad programming style. Use parentheses to make the computation easier to read.

```
2^5;
exp(4);
ifactor(3^27);
```

Maple knows standard mathematical notation. The function `exp(x)` is used for  $e^x$ . The command `ifactor` does integer factoring; the command `factor` factors polynomials.

```
12345/98765;
evalf(12345/98765);
```

You can force Maple to approximate values as 'floating point numbers' (i.e., using decimal notation). Some of Maple's commands do this automatically, as we will see later.

```
I^2;
evalf(Pi);
evalf(exp(1));
```

The number  $i = \sqrt{-1}$  is denoted by  $I$ ; the case is significant—a lower case 'i' is different.  $Pi$  denotes  $\pi$ ; again case is significant. To get the constant  $e$ , use `exp(1)`.

### 1.2.2 Assignment versus Equality

One of the strengths of a CAS is to store certain objects (which may have a lengthy algebraic description) to a variable. The general format for assigning  $B$  to variable  $A$  is:

$$A := B$$

For example,

```
F:=(x+2)^2;  
expand(F);  
subs(x=3,F);
```

We put the object  $(x + 2)^2$  into the variable  $F$ . We can then expand the object, evaluate at certain values of  $x$ , and so on.

```
B:= 5*x+3=-4;  
solve(B,x);
```

The equal sign is used in many places. For example, an object can be an equation that we store in the variable  $B$ , then we can solve that equation for  $x$ .

### 1.2.3 Hints about Expressions

Maple can manipulate expressions involving variables as well as numbers. Variables can be assigned values, either actual numeric values or entire expressions.

```
F:=(x+5)^2;  
x:=5;  
F;  
x^2;
```

This is not a preferred way of putting the value 5 into the expression for  $F$ , because it sets  $x$  permanently to 5.

```
x:='x';  
x;  
F;
```

#### How to clear a variable.

This clears the value of  $x$ —this is not the same as setting it to zero! Note that Maple remembers the ‘real’ definition of  $F$ —it is still an expression involving  $x$ , not the fixed value 100.

In expressions, always be explicit with multiplication signs.

```
expand(F);  
factor(F);  
factor(x^2+4*x+5);
```

If a polynomial doesn’t factor, Maple leaves it alone. (As a general rule, if Maple cannot perform a calculation it will simply restate the command.)

**Good Editing Note:** You can go back anywhere in your Maple session and change a command. When you do this you change *only* the result of that command—you do not change any subsequent commands that depend on the result. If you want to re-execute subsequent commands, you must move the cursor to each command in turn and press the return key. Do not overuse this editing ability—it can make the chain of events very difficult to follow. It often is better to copy the command, paste it in at the end of the session, modify it there, and execute it as a new command.

### 1.2.4 Evaluation and Substitution

It usually is not a good idea to assign values to variables (like:  $x:=3$ ;)—they are more useful as variables. If you want Maple to evaluate an expression you can give the variables temporary values.

```
f:=A*(x+3)^3;  
subs(A=3,f);  
subs({x=5,A=2},f);  
x; A; f;
```

Note here that you use a plain ‘=’ inside the substitution command. Two or more substitutions are done by enclosing them in curly braces. The values of  $f$ ,  $A$  and  $x$  *don’t* change.

```

subs(x=z^2,f);
f:=subs(x=z^3,f);
f;

f:=x^2+3*x+4;
f(1);
g:=x -> x^2+3*x+4;
g(1);
h:=unapply(f,x);
h(1);
h(x);
h(x^3);

map(g,[-3,-2,-1,0,1,2,3]);
map(g,{-3,-2,-1,0,1,2,3});

```

You can substitute whole expressions for variables, not just values. To force the value of  $f$  to change, just reassign it—but be careful. It usually is better to give the expression a new name in case you need the original definition of  $f$  later.

Maple makes a crucial distinction between an expression like  $x^2 + 3x + 4$  and the function which assigns to each value of  $x$  the value  $x^2 + 3x + 4$ . In this example,  $f$  is an **expression**,  $g$  is a **function**. We also show how to convert an expression into a function, that's  $h$ . Finally, we show that we can evaluate  $h$  at a number or an expression.

Using the ‘map’ operation and a function (not an expression) you can evaluate a function on a list of values. The list can be enclosed in square brackets or in set braces—but notice the difference! The set braces produce a set as a result, meaning that repeated values are shown only once, and the order does not necessarily correspond to the order of the input values. In general, the square bracket form is more useful.

### 1.2.5 Solving Equations

Maple will solve a limited class of equations exactly and a broader class approximately.

```

solve(x^2-x-10=10,x);
solutions:=solve(g(x),x);
solve(x^5+4*x^3-3*x^2+10,x);
solve(tan(x)=x+1,x);
allvalues(%);
solve(tan(x)=x,x);
allvalues(%);
solve(sin(x)=cos(x)-1,x);

```

The *solve* command will give exact answers if possible, including complex answers. If Maple can't solve an equation it may print nothing, or it may display the equation in a standard form, preceded by ‘RootOf’. You can ask Maple to attempt to evaluate roots displayed as a RootOf using *allvalues*. Maple will attempt to find all solutions for an equation or a RootOf, but often will not succeed.

Do not rely on the *solve* command alone when looking for solutions to equations—at the very least, you should graph your functions so that you know where roots appear to be. Note that if you give an expression instead of an equation Maple assumes that you mean ‘= 0’.

```

map(g,[solutions]);
simplify(%);

```

In some cases, Maple will not automatically simplify as much as it can, so you need to nudge it a bit. In this case, of course, we knew that the values should be zero. If the result had not simplified to zero, careful scrutiny of the problem would be called for.



```
f:=x^3+3*x^2+x+5:
solve(f,x);
fsolve(f,x);
fsolve(f,x,complex);
```

You can end a line with a colon instead of a semicolon—it tells Maple not to display the result. This can be particularly handy when the result is very long, or simply when you either know what it is or do not need to know (as when you will be using the result to compute a further result). The *fsolve* operation tells Maple to approximate the roots. In this example, Maple can compute the roots exactly, but it may be more informative to see approximations. Note Maple only computes real roots unless the ‘complex’ option is present. If the function is not a polynomial, *fsolve* will find at most one solution, but for polynomials it will find all solutions.

```
g:=x -> exp(x)+x^2-10*x;
fsolve(g(x),x);
fsolve(g(x),x,-2..2);
g(%);
fsolve(g(x),x,2..4);
g(%);
```

You can specify a range in *fsolve*—you would normally do this when you know there are multiple roots and you want to approximate a particular one. In this case, a graph of the function clearly indicates a root near 0 and a root near 3. Because *g* is a function, and *fsolve* expects an expression, it is necessary to use *g(x)* in *fsolve*. (I made *g* a function so that it would be easy to check the answers.)

```
solve({3*x+5*y=6,4*x-6*y=-8},
      {x,y});
h1:=3*x-6*y-5;
h2:=-4*x*y+7;
solve({h1,h2},{x,y});
evalf(%);
fsolve({h1,h2},{x,y});
fsolve({h1,h2},{x,y},{x=0..6,y=0..2});
```

Maple can solve equations simultaneously, with either *solve* or *fsolve*. As with single equations, *fsolve* may not find all solutions, but you can specify ranges for *x* and *y* to force *fsolve* to find particular solutions.

### 1.3 Reading Questions

At this point, you should be able to answer the following questions. If you’re not sure, go back through the examples on the previous couple of pages to track down the solutions.

1. What is the Maple command to compute:
  - (a)  $e^3$ ?
  - (b)  $\sin(\pi/4)$ ?
  - (c) the integer factors for 255?
2. What Maple command will give a numerical approximation to  $\sqrt{1245}$ ?
3. What Maple commands would:
  - (a) Assign the expression  $6x^2 + x - 1$  to the variable name: *PolyName*

- (b) Factor the previous expression.
- (c) Evaluate the expression at the point  $x = 1.23$ .
- (d) Solve for where the expression is equal to zero.
- (e) Convert the expression into a function.

## Solutions

1. What is the Maple command to compute:
  - (a)  $e^3$ ? SOLUTION: `exp(3)`
  - (b)  $\sin(\pi/4)$ ? SOLUTION: `sin(Pi/4)` (Note the uppercase  $P$ !)
  - (c) the integer factors for 255? SOLUTION: `ifactor(255)`
2. What Maple command will give a numerical approximation to  $\sqrt{1245}$ ?  
SOLUTION: `evalf(\sqrt{1245})`
3. What Maple commands would:
  - (a) Assign the expression  $6x^2 + x - 1$  to the variable name: *PolyName*  
`PolyName:=6*x^2+x-1`
  - (b) Factor the previous expression.  
`factor(PolyName)`
  - (c) Evaluate the expression at the point  $x = 1.23$ .  
`subs(x=1.23, PolyName)`
  - (d) Solve for where the expression is equal to zero.  
`solve(PolyName=0,x);`
  - (e) Convert the expression into a function.  
`g:=unapply(PolyName,x);`

## 1.4 Plotting

Maple can do two and three dimensional graphing. There are many ways to adjust the plots that Maple produces; we'll cover just the basics here.

```
plot(f,x=-10..10);
plot(g(x),x=-5..5);
plot(g,-10..10);
plot(g,-10..10,y=-10..10);
```

Maple will accept either an expression or a function to plot, but notice that the form of the range is a bit different in the two cases. For an expression, use 'x=' before the range, but for a function leave it out. You can also specify the range for  $y$  explicitly. This is often useful if the function gets very large on the interval.

You can change the appearance of the plot in a variety of ways. Click in the plot and you should see a border appear around it; then you can explore the items on the plot menu. If you prefer, you can have plots come up in separate windows; use the **Options** item on the **Tools** menu in the main Maple window, then open the Display tab and change the plot display.

```
plot([1+2*sin(t),t,t=-Pi..Pi],
     coords=polar);
```

You can do plots in polar coordinates. Inside the square brackets, the first expression is the radius, the second is the angle, and the third is the range of the variable. In this case, the angle and the variable are the same, but in general the angle could be a more complicated function of the variable.

### Multiple Plots in One Graph

```
plot({x^2,x^3},x=-5..5);
plot({
  [1+2*sin(t),t,t=-Pi..Pi],
  [cos(4*t),t,t=-Pi..Pi]},
     coords=polar);
```

You can put more than one curve in the same plot by enclosing a list of expressions in curly braces.

```
with(plots):
polarplot(1+sin(t));
polarplot(sin(t/3));
polarplot(sin(t/3),
           t=0..3*Pi);
```

When you start Maple, only the most basic plotting commands are available. By typing “with(plots)” you tell Maple to load a number of special purpose plotting functions. One of the most useful is the polarplot routine shown here. By default, Maple uses the range  $[-\pi, \pi]$ ; the last command shows how to specify a different range.

## 1.5 Calculus

In this section, we look at basic differentiation and integration. Later we’ll look at solving differential equations.

### 1.5.1 Differentiation

```
f:=x->x^3+4*x^2+3*x+4:
fp:=diff(f(x),x);
G:=x*sin(x)/(x^2+4);
diff(G,x);
simplify(diff(G,x));
```

The *diff* command is what we use. Notice that the first argument you give is an *expression*, and secondly, the name of the variable. In the second example, you might want to simplify the expression.

```
G:=x*sin(x)-exp(x)*cos(x);
dG:=diff(G,x);
d2G:=diff(dG,x);
ddG:=diff(G,x$2);
```

The second derivative can be performed two ways- take the derivative of the derivative, or directly (in the line ddG).

```
G:=x*sin(y)-exp(y)*cos(x);
dGx:=diff(G,x);
dGy:=diff(G,y);
dGxx:=diff(G,x$2);
dGxy:=diff(G,y,x,y);
```

The command `diff` works in a similar way for multivariate expressions. Note the last couple of commands tells us how to compute a higher order derivative- In this case  $G_{xx}$  and  $G_{yxy}$

## 1.5.2 Integration

```
f:=(2*z-3)*sqrt(z-3);
g:=int(f,z);
diff(%,z);
factor(%)
```

The `int` command shown here will compute an antiderivative. The computer is not infallible, and you may not notice a typing mistake, so it's not a bad idea to use Maple to check your work. Here, I've taken the derivative to check the antiderivative.

```
A:=int(f,z=5..10);
B:=simplify(A);
C:=evalf(A);
subs(z=10,g)-subs(z=5,g);
```

To do definite integration give the limits after the variable. Of course, you can also do it yourself using the antiderivative. Notice that the answer can look quite different before simplification.

```
f:=1/sqrt(x):
int(f,x=0..1);
f:=1/x^2:
int(f,x=0..1);
int(f,x=1..infinity);
f:=exp(-x^2):
int(f,x=-infinity..infinity);
int(sin(x),x=0..infinity);
```

Maple can do many improper integrals exactly. If the integral does not converge, Maple may answer with ' $\infty$ ' if appropriate, or it may give an error message.

```
f:=tan(x^2):
int(f,x);
A:=int(f,x=0..1);
evalf(A);
B:=Int(f,x=0..1);
evalf(B);
```

Maple can't find antiderivatives of some functions. In such cases you can use `evalf` to approximate the value of a definite integral. If you don't want Maple to even try to find an antiderivative, use the capital I form of the integration operator.

## Reading Questions

Try answering the following questions after working through the previous Maple exercises.

1. If  $f(x, y) = x^2 + xy - 5y^2 + \sin(y)$ , have Maple compute  $f_{xy}(3)$ .
2. If  $f(x) = xe^{-x^2}$ , then have Maple plot the derivative and an antiderivative of  $f$  with all three functions on the same plot. Make the interval  $-3 \leq x \leq 3$ .
3. Is the following accurate in Maple? If not, identify the issue and correct it:

```
g:=3x^2+5*x+7;
diff(g(x),x);
```

## Chapter 2

# Working with Worksheets

Now that we've gotten a little experience with typing in Maple commands, let's take a quick look at editing and adding/deleting execution groups in the worksheet.

First, if we go to: **Insert** -> **Execution Group** and then choose either before of after your cursor, you can add groups into your worksheet.

### 2.1 Adding text to a Maple session

You can add plain text to a *Maple* session. This is especially useful if you are going to give a printed or electronic version of a session to someone else, since it lets you explain what is going on. It is also a good idea to add text to remind yourself at a later date what you were doing and why.

To add text above a command, position the cursor anywhere in the command. Then open the **Insert** menu, and then the **Paragraph** submenu, and choose **Before Cursor**. When you type you will get ordinary text that will not be interpreted as part of the command.

### 2.2 Saving your work

You can save your Maple worksheet so that you can come back to it at a later time. Choose the **Save** or **Save As...** item from the **File** menu in Maple; a new window will pop up. (Use **Save As** if you want to save your worksheet under a new name.) You may also save a worksheet using the third button from the left in the menu bar.

When you reload a worksheet (with the **Open** item on the **File** menu or with the second button from the left) Maple will not automatically recreate all the values created in the worksheet. You may selectively re-execute commands in the worksheet, or you may use the **Execute** item on the **Edit** menu to execute the entire worksheet.

*HINT:* The size of the file is a lot smaller if you save only your Maple commands and leave out plots, animations, and Maple output. To remove the output before saving your file, go to: **Edit**, then **Remove Output**, then **From Worksheet**. Once this is saved, if you want all the commands to be re-executed after opening the file, select the triple exclamation point button: **!!!**.

## 2.3 Printing

**BEFORE YOU PRINT:** Please clean up your session before you print, so that you do not waste paper and so that the printed copy is attractive and easy to read. If you want to print only part of the session, first save the worksheet, and then cut out the portions of the session that you do not want to print. You might want to save the trimmed version using a different name.

To print your worksheet, choose **Print** from the **File** menu, or click on the print button (the fourth little button from the left in the menu bar).

If you would like to print just a plot, you need to display it in a window of its own. You can do this by changing the plot display option, as described in the *Plotting* section. Then position the cursor in the plot command and press return. The plot should open in a new window, and you can use the Print menu in that window to print the plot.

Its not very often that you should need to print an actual hard copy. Notice that you can also **Export** the worksheet to a PDF file, which is what you would typically turn in, together with the actual Maple file (the filename ends in *.mw*).

## Chapter 3

# Lab 1: Slope Fields

In this lab, we want to look at how to define differential equations in Maple, then we'll look at graphical solution techniques including the slope field.

Defining a differential equation is straightforward in Maple. Here are a couple of examples:

$$\begin{array}{ll} y' = 3y + t & \text{becomes } \text{diff}(y(t), t) = 3*y(t) + t \\ y'' = 3y' - 3y + x & \text{becomes } \text{diff}(y(x), x^2) = 3*\text{diff}(y(x), x) - 3*y(x) + x \end{array}$$

In each case, note that we have to be explicit about what the dependent and independent variables are- this is important!

When we actually define a DE in Maple, it is often convenient to assign it a variable name- for example, storing it as the variable  $A$ :

```
A:=diff(y(x),x)=(3*x+y(x))/(y(x)-2);
```

Now we only need to write  $A$  in place of the DE.

### EXAMPLE 1:

Plot the direction field for

$$y' = \frac{x^2}{1 - y^2}$$

First, define the differential equation and assign it to a variable (in this case, `diffeqn`). In the first example, we treat  $y(x)$  as an expression in  $x$ :

```
A:=diff(y(x),x)=x^2/(1-(y(x))^2);
```

Alternatively, you can define it in terms of functions. In this case, we use the uppercase  $D$  form of the derivative:

```
B:=D(y)(x)=x^2/(1-(y(x))^2);
```

And now we can plot the direction field as either:

```
with(DEtools):
DEplot(A,y(x),x=-4..4,y=-4..4);
DEplot(B,y(x),x=-4..4,y=-4..4);
```




Figure 3.1: Solution to Example 1, the direction field for  $y' = x^2/(1 - y^2)$ .

In either event, you should get the slope field in Figure 3.1. The arrows are pointing in the direction of the solution curve.

We'll now take a look at some optional commands. Try redoing the last line, and make a note at how the plot changes:

```
DEplot(A,y(x),x=-4..4,y=-4..4, dirgrid=[30,30],title="First Plot");
```

We can use the same command to plot some sample solution curves along with the direction field. To do this, we will have to include some initial values.

## EXAMPLE 2

Plot the direction field for the given differential equation, together with the three sample solutions through  $(1, 2)$ ,  $(1, 0)$  and  $(1, 1)$ , if

$$ty' + 2y = 4t^2$$

First, define the differential equation, then plot:

```
B:=t*diff(y(t),t)+2*y(t)=4*t^2;
DEplot(B,y(t),t=0.01..2,[[y(1)=2],[y(1)=0],[y(1)=1]],y=-2..4);
```

See Figure 3.2 for the solution.

## 3.1 Lab 1 Questions

1. Plot the following direction fields:

(a) The form for these is:  $y' = f(t, y)$






Figure 3.2: Solution to Example 2, the direction field for  $ty' + 2y = 4t^2$  with three solution curves.

- i.  $y' = 1 + 3 \sin(t) + y$
  - ii.  $y' = \frac{1}{2}y + 2 \cos(t)$
- (b) The form for these is:  $y' = f(y)$ . These equations are known as *autonomous equations*, since the differential equation doesn't depend explicitly on time.
- i.  $y' = y(3 - y)$
  - ii.  $y' = y(y - 1)(y - 2)$
- (c) The form for these is:  $y' = f(t)$
- i.  $y' = 3t^2 - 2$
  - ii.  $y' = \frac{\sin(t)}{t}$

Analyze your output from this section. Given a direction field, is it possible to determine the form of the differential equation (that is, is it  $y' = f(y)$ ,  $y' = f(t)$ , or something else)?

2. In Problem 1(b), is it possible to determine the behavior of the solution,  $y$ , as  $t \rightarrow \infty$ ? Describe this behavior in terms of its dependence on the initial value,  $y(0)$ .
3. Let

$$y' = 3 \sin(t) + 1 + y$$

Based on the direction field, determine the behavior of  $y$  as  $t \rightarrow \infty$ . If this behavior depends on the initial value of  $y$  at  $t = 0$ , describe this dependency.

Think about this question and I'll give you a new command next week!